
Differential Forms Using xTerior

Downloading and Installing xAct (xTerior is included)

Visit the page: <http://www.xact.es>

Follow installation instructions on <http://www.xact.es/download.html>

Linux:

1. Download the tarball xAct_V.tgz (V is the version number)
2. `sudo -i ; cd /usr/share/Mathematica/Applications/; tar xvfz ~/Downloads/xAct_V.tgz`

Windows:

1. Download the zip file xAct_V.zip (V is the version number)
2. unzip its contents in C:\Program Files\Wolfram Research\Mathematica\<version>\AddOns\Applications\

Read the documentation: <http://www.xact.es/xTerior/index.html>

<http://www.xact.es/Documentation/English/xTeriorDoc.nb>

Linux: /usr/share/Mathematica/Applications/xAct/Documentation/English/xTeriorSlideShowYT.nb

Windows: C:\Program Files\Wolfram Research\Mathematica\<version>\AddOns\Applications\xAct\Documentation\English

Start an xTerior session

If you are already running a Mathematica session, esp. if you have loaded xTerior, xAct, ..., make a call to `Quit[]` before the loading of the package:

```
Quit[]; Needs["xAct`xTerior`"]
```

```
In[ ]:= Needs["xAct`xTerior`"]
```

Introduction

First we define a manifold of symbolic dimension `dim` and the symbols used for its indices:

```
In[ ]:= DefConstantSymbol[dim, PrintAs → "D"];
DefManifold[M, dim, {μ, ν, μ1, μ2, ν1, ν2}];
```

** DefConstantSymbol: Defining constant symbol dim.

Then we define forms of different types: a 0-form is a scalar

```
In[ ]:= DefDiffForm[f[], M, 0, PrintAs → "f"];
DefDiffForm[σ[], M, 1, PrintAs → "σ"];
DefDiffForm[λ[], M, 2, PrintAs → "λ"];
DefDiffForm[ω[], M, 3, PrintAs → "ω"];
```

** DefTensor: Defining tensor f[].

The wedge product uses the \wedge character: typeset with [Esc]^ [Esc].

Notice that f is a scalar, so the wedge product is reduced to ordinary multiplication

```
In[ ]:= f[] ∧ σ[]
```

```
Out[ ]:= f σ
```

No simplification of the expression:

```
In[ ]:= λ[] ∧ σ[] + σ[] ∧ λ[]
```

```
Out[ ]:= λ ∧ σ + σ ∧ λ
```

Use the ToCanonical function to make simplifications and write an expression in canonical form when wedge products are present:

```
In[ ]:= λ[] ∧ σ[] + σ[] ∧ λ[] // ToCanonical
```

```
Out[ ]:= 2 λ ∧ σ
```

```
In[ ]:= σ[] ∧ ω[] + ω[] ∧ σ[] // ToCanonical
```

```
Out[ ]:= 0
```

The Exterior derivative is Diff[x], x an expression

```
In[ ]:= Diff[σ[] ∧ λ[]]
```

```
Out[ ]:= d[σ] ∧ λ - σ ∧ d[λ]
```

```
In[ ]:= Diff[f[] σ[]] ∧ Diff[λ[]] ∧ Diff[ω[]]
```

```
Out[ ]:= f d[σ] ∧ d[λ] ∧ d[ω] + d[f] ∧ σ ∧ d[λ] ∧ d[ω]
```

Define a manifold of specific dimension (here 3) and a chart with coordinates (x, y, z):

```
In[ ]:= DefManifold[M3, 3, {α, β, γ, δ}];
DefChart[cartesian, M3, {1, 2, 3}, {x[], y[], z[]}];
```

** DefManifold: Defining manifold M3.

Then we can use the basis 1-forms {dx, dy, dz} to write down differential forms:

```
In[ ]:= {Diff[x[]], Diff[y[]], Diff[z[]]}
```

```
Out[ ]:= {d[x], d[y], d[z]}
```

```
In[ ]:= ω2 = (x[] ^ 2 + y[] ^ 2) Diff[z[]] ∧ Diff[x[]] - 1 / (x[] ^ 2 + y[] ^ 2) Diff[x[]] ∧ Diff[y[]]
```

```
Out[ ]:= -  $\frac{d[x] \wedge d[y]}{x^2 + y^2} + d[z] \wedge d[x] (x^2 + y^2)$ 
```

Notice that the $dz \wedge dx \wedge dz$ term is not set to 0:

```
In[ ]:= ω2 ∧ Diff[z[]]
```

```
Out[ ]:= -  $\frac{d[x] \wedge d[y] \wedge d[z]}{x^2 + y^2} + d[z] \wedge d[x] \wedge d[z] (x^2 + y^2)$ 
```

Simplification[expr] is a wrapper to Simplify[ToCanonical[expr]]. It makes $dz \wedge dx \wedge dz$ to vanish.

```
In[ ]:= ω2 ∧ Diff[z[]] // Simplification
```

```
Out[ ]:= -  $\frac{d[x] \wedge d[y] \wedge d[z]}{x^2 + y^2}$ 
```

```
In[ ]:= Diff[ω2] // Simplification
```

```
Out[ ]:= 2 d[x] ∧ d[y] ∧ d[z] y
```

We can also work with scalars. A scalar function is a head that can take as its arguments any number of coordinates.

Remember that \wedge (UpSet) assigns a symbolic property of a symbol:
 PrintAs[xx] \wedge = "x"; area[circle] = $2\pi r^2$; rate[euro] = 1.1; rate[usd] = 1.0;

```
In[ ]:= DefScalarFunction !@ {ψx, ψy, ψz};
PrintAs@ψx  $\wedge$  = "ψx";
PrintAs@ψy  $\wedge$  = "ψy";
PrintAs@ψz  $\wedge$  = "ψz";
```

** DefScalarFunction: Defining scalar function ψx.

```
In[*]:=  $\psi = \psi_x[x[], y[], z[]] \text{Diff}[y[]] \wedge \text{Diff}[z[]] +$   

 $\psi_y[x[], y[], z[]] \text{Diff}[z[]] \wedge \text{Diff}[x[]] +$   

 $\psi_z[x[], y[], z[]] \text{Diff}[x[]] \wedge \text{Diff}[y[]]$ 
```

```
Out[*]:=  $d[y] \wedge d[z] \psi_x[x, y, z] + d[z] \wedge d[x] \psi_y[x, y, z] + d[x] \wedge d[y] \psi_z[x, y, z]$ 
```

Divergence $\nabla \cdot \psi$

$$d\psi = (\partial_x \psi_x + \partial_y \psi_y + \partial_z \psi_z) dx \wedge dy \wedge dz$$

```
In[*]:= Diff[ψ] // Simplification
```

```
Out[*]:=  $d[x] \wedge d[y] \wedge d[z] (\psi_z^{(0,0,1)}[x, y, z] + \psi_y^{(0,1,0)}[x, y, z] + \psi_x^{(1,0,0)}[x, y, z])$ 
```

```
In[*]:=  $\phi = \psi_x[x[], y[], z[]] \text{Diff}[x[]] +$   

 $\psi_y[x[], y[], z[]] \text{Diff}[y[]] +$   

 $\psi_z[x[], y[], z[]] \text{Diff}[z[]]$ 
```

```
Out[*]:=  $d[x] \psi_x[x, y, z] + d[y] \psi_y[x, y, z] + d[z] \psi_z[x, y, z]$ 
```

The curl $\nabla \times \psi$

$$d\phi = (\partial_y \psi_z - \partial_z \psi_y) dy \wedge dz + (\partial_x \psi_z - \partial_z \psi_x) dx \wedge dz + (\partial_x \psi_y - \partial_y \psi_x) dx \wedge dy$$

```
In[*]:= Diff[φ] // Simplification
```

```
Out[*]:=  $d[y] \wedge d[z] (-\psi_y^{(0,0,1)}[x, y, z] + \psi_z^{(0,1,0)}[x, y, z]) +$   

 $d[x] \wedge d[y] (-\psi_x^{(0,1,0)}[x, y, z] + \psi_y^{(1,0,0)}[x, y, z]) + d[x] \wedge d[z] (-\psi_x^{(0,0,1)}[x, y, z] + \psi_z^{(1,0,0)}[x, y, z])$ 
```

Change coordinates: This defines a chart with the spherical coordinates. We have also defined cartesian to be the Cartesian coordinates on M3:

```
In[*]:= DefChart[spherical, M3, {1, 2, 3}, {r[], θ[], φ[]}]
```

```
** DefChart: Defining chart spherical.
```

Define coordinate transformation:

```
In[*]:= cartesian2spherical = {  

 $x[] \rightarrow r[] \text{Sin}[\theta[]] \text{Cos}[\phi[]],$   

 $y[] \rightarrow r[] \text{Sin}[\theta[]] \text{Sin}[\phi[]],$   

 $z[] \rightarrow r[] \text{Cos}[\theta[]]$   

}
```

```
Out[*]:=  $\{x \rightarrow \text{Cos}[\phi] r \text{Sin}[\theta], y \rightarrow r \text{Sin}[\theta] \text{Sin}[\phi], z \rightarrow \text{Cos}[\theta] r\}$ 
```

In[]:= $\omega_2 = (x[]^2 + y[]^2) \text{Diff}[z[]] \wedge \text{Diff}[x[]] - 1 / (x[]^2 + y[]^2) \text{Diff}[x[]] \wedge \text{Diff}[y[]]$

Out[]:=
$$-\frac{d[x] \wedge d[y]}{x^2 + y^2} + d[z] \wedge d[x] (x^2 + y^2)$$

In[]:= $\omega_2 /. \text{cartesian2spherical} // \text{Simplification}$

Out[]:=
$$\text{Cos}[\varphi] r^3 \text{Sin}[\theta]^2 d[r] \wedge d[\theta] - \frac{(1 + \text{Cos}[\theta] r^4 \text{Sin}[\theta]^3 \text{Sin}[\varphi]) d[r] \wedge d[\varphi]}{r} + (-\text{Cot}[\theta] + r^4 \text{Sin}[\theta]^4 \text{Sin}[\varphi]) d[\theta] \wedge d[\varphi]$$

Volume 3-form in spherical coordinates:

In[]:= $\epsilon = \text{Diff}[x[]] \wedge \text{Diff}[y[]] \wedge \text{Diff}[z[]] /. \text{cartesian2spherical} // \text{Simplification}$

Out[]:=
$$r^2 \text{Sin}[\theta] d[r] \wedge d[\theta] \wedge d[\varphi]$$

In[]:= $\omega = r[]^2 \text{Sin}[\theta[]] \text{Diff}[r[]] \wedge \text{Diff}[\theta[]] \wedge \text{Diff}[\varphi[]]$

Out[]:=
$$r^2 \text{Sin}[\theta] d[r] \wedge d[\theta] \wedge d[\varphi]$$

Integrate: First remove the head Wedge, then perform ordinary integration:

FullForm[ω]

$\omega // \text{Head}$

In[]:= $\omega /. _Wedge \rightarrow 1$

Out[]:=
$$r^2 \text{Sin}[\theta]$$

In[]:= **Integrate[$\omega /. _Wedge \rightarrow 1$, { θ [], 0, π }, { φ [], 0, 2π }]**

Out[]:=
$$4 \pi r^2$$

Schwarzschild Geometry

The Schwarzschild metric in (t, r, θ, φ) coordinates is:

$$g = -\left(1 - \frac{2M}{r}\right)^{-1} dt \otimes dt + \left(1 - \frac{2M}{r}\right) dr \otimes dr + r^2 d\theta \otimes d\theta + r^2 \sin^2(\theta) d\varphi \otimes d\varphi$$

Its determinant is:

$$\sqrt{-g} = r^2 \sin(\theta)$$

and the Levi-Civita tensor:

$$\epsilon = \sqrt{-g} dt \wedge dr \wedge d\theta \wedge d\varphi$$

Here, we quit the kernel to start a new session with all symbols undefined, we use the `Quit[]` function. We have to reload `xTerior`:

```
In[*]:= Quit[]
```

```
In[*]:= Needs["xAct`xTerior`"]
```

```
DefManifold[M4, 4, {μ, ν, α, β, γ}];
DefChart[trθφ, M4, {0, 1, 2, 3}, {t[], r[], θ[], φ[]];
DefChart[Kruskal, M4, {0, 1, 2, 3}, {U[], V[], θ[], φ[]];

DefConstantSymbol[M, PrintAs -> "M"]
```

** DefManifold: Defining manifold M4.

```
In[*]:= DefDiffForm[ε[], M4, 4, PrintAs -> "ε"]
```

** DefTensor: Defining tensor ε[].

```
In[*]:= ε[] = r[]^2 Sin[θ[]] Diff[t[]] ∧ Diff[r[]] ∧ Diff[θ[]] ∧ Diff[φ[]]
```

```
Out[*]:= r^2 Sin[θ] d[t] ∧ d[r] ∧ d[θ] ∧ d[φ]
```

Coordinate Transformation, $r > 2M$:

$$U = \sqrt{\frac{r}{2M} - 1} e^{\frac{r}{4M}} \cosh\left(\frac{t}{4M}\right), V = \sqrt{\frac{r}{2M} - 1} e^{\frac{r}{4M}} \sinh\left(\frac{t}{4M}\right)$$

```
In[*]:= Kruskal2trθφ = {
```

$$U[] \rightarrow \sqrt{\frac{r[]}{2M} - 1} e^{\frac{r[]}{4M}} \text{Cosh}\left[\frac{t[]}{4M}\right],$$

$$V[] \rightarrow \sqrt{\frac{r[]}{2M} - 1} e^{\frac{r[]}{4M}} \text{Sinh}\left[\frac{t[]}{4M}\right]$$

```
}
```

```
Out[*]:= {U -> e^{r/4M} Cosh[t/4M] sqrt[-1 + r/2M], V -> e^{r/4M} sqrt[-1 + r/2M] Sinh[t/4M]}
```

```
In[ ]:= {Diff[U[]] /. Kruskal2trθφ, Diff[U[]] ∧ Diff[V[]] /. Kruskal2trθφ} // Simplification
```

$$\text{Out[]:= } \left\{ \frac{e^{\frac{r}{4M}} \left(\text{Cosh}\left[\frac{t}{4M}\right] d[r] r + d[t] (-2M + r) \text{Sinh}\left[\frac{t}{4M}\right] \right)}{4M^2 \sqrt{-4 + \frac{2r}{M}}}, \frac{e^{\frac{r}{2M}} r d[r] \wedge d[t]}{32M^3} \right\}$$

The metric in the Kruskal coordinates is:

$$\mathbf{g} = \frac{32M^3}{r} e^{\frac{-r}{2M}} (-dV \otimes dV + dU \otimes dU) + r^2 d\theta \otimes d\theta + r^2 \sin^2(\theta) d\varphi \otimes d\varphi$$

so the determinant is

$$-g = \left(\frac{32M^3}{r} e^{\frac{-r}{2M}} \right)^2 r^2 r^2 \sin^2(\theta)$$

We compute the Levi-Civita symbol in the new coordinates and then transform back to spherical. The result is the same as the ϵ computed above:

```
In[ ]:= sqrtg = 32 M^3 e^{\frac{-r[]}{2M}} r[] Sin[θ[]]; (*that is √-g*)
ε[] = sqrtg Diff[U[]] ∧ Diff[V[]] ∧ Diff[θ[]] ∧ Diff[φ[]]
```

$$\text{Out[]:= } 32 e^{\frac{-r}{2M}} M^3 r \text{Sin}[\theta] d[U] \wedge d[V] \wedge d[\theta] \wedge d[\varphi]$$

```
In[ ]:= ε[] /. Kruskal2trθφ // Simplification
```

$$\text{Out[]:= } r^2 \text{Sin}[\theta] d[r] \wedge d[t] \wedge d[\theta] \wedge d[\varphi]$$

Simplify notation for exterior derivative:

```
In[ ]:= d = Diff
```

```
In[ ]:= d[ε[]]
```

```
In[ ]:= sqrtg d[U[]] ∧ d[V[]] ∧ d[θ[]] ∧ d[φ[]]
```

Simplify notation

```
In[ ]:= Quit[]
```

```
In[ ]:= Needs["xAct`xTerior`"]
```

```
In[*]:= DefManifold[      M4, 4, {μ, ν, α, β, γ}];
DefChart[trθφ      , M4, {0, 1, 2, 3}, {T[], R[], Θ[], Φ[]};
DefChart[Kruskal, M4, {0, 1, 2, 3}, {U[], V[], Θ[], Φ[]};

DefConstantSymbol[M, PrintAs → "M"]
```

** DefManifold: Defining manifold M4.

Translate symbols to ones that are easier to use and read, without too many brackets.

Notice the use of PrintAs, so that we see expressions in the more familiar form.

```
In[*]:= t = T[]; dt = Diff[T[]]; PrintAs@T ^= "t";
r = R[]; dr = Diff[R[]]; PrintAs@R ^= "r";
θ = Θ[]; dθ = Diff[Θ[]]; PrintAs@Θ ^= "θ";
φ = Φ[]; dφ = Diff[Φ[]]; PrintAs@Φ ^= "φ";

u = U[]; du = Diff[U[]]; PrintAs@U ^= "u";
v = V[]; dv = Diff[V[]]; PrintAs@V ^= "v";
```

```
In[*]:= kruskal2trθφ = {
  u → √(r/(2M) - 1) e^(r/(4M)) Cosh[t/(4M)],
  v → √(r/(2M) - 1) e^(r/(4M)) Sinh[t/(4M)]
}
```

```
Out[*]:= {u → e^(r/(4M)) Cosh[t/(4M)] √(-1 + r/(2M)), v → e^(r/(4M)) √(-1 + r/(2M)) Sinh[t/(4M)]}
```

```
In[*]:= DefDiffForm[ε[], M4, 4, PrintAs → "ε"];
```

```
In[*]:= sqrtg = 32 M^3 e^(-r/(2M)) r Sin[θ]; (*that is √-g*)
ε[] = sqrtg du ∧ dv ∧ dθ ∧ dφ
```

```
Out[*]:= 32 e^(-r/(2M)) M^3 r Sin[θ] d[u] ∧ d[v] ∧ d[θ] ∧ d[φ]
```


In[]:= $\epsilon[]$ /. kruskal2tr $\theta\phi$ // Simplification

Out[]:= $r^2 \sin[\theta] d[r] \wedge d[t] \wedge d[\theta] \wedge d[\phi]$

Various forms:

In[]:= {du, dv, du \wedge dv, Diff[u dv]}

Out[]:= {d[u], d[v], d[u] \wedge d[v], d[u] \wedge d[v]}

In[]:= {du, dv, du \wedge dv} /. kruskal2tr $\theta\phi$ // Simplification

Out[]:=
$$\left\{ \frac{e^{\frac{r}{4M}} \left(\cosh\left[\frac{t}{4M}\right] d[r] r + d[t] (-2M + r) \sinh\left[\frac{t}{4M}\right] \right)}{4M^2 \sqrt{-4 + \frac{2r}{M}}}, \right.$$

$$\left. \frac{e^{\frac{r}{4M}} \left(-\cosh\left[\frac{t}{4M}\right] d[t] (2M - r) + d[r] r \sinh\left[\frac{t}{4M}\right] \right)}{4M^2 \sqrt{-4 + \frac{2r}{M}}}, \frac{e^{\frac{r}{2M}} r d[r] \wedge d[t]}{32M^3} \right\}$$

Since $U = \sqrt{\frac{r}{2M} - 1} e^{\frac{r}{4M}} \cosh\left(\frac{t}{4M}\right)$, we should obtain the same result as for du above:

In[]:= Diff[$\sqrt{\frac{r}{2M} - 1} e^{\frac{r}{4M}} \cosh\left[\frac{t}{4M}\right]$] // Simplification

Out[]:=
$$\frac{e^{\frac{r}{4M}} \left(\cosh\left[\frac{t}{4M}\right] d[r] r + d[t] (-2M + r) \sinh\left[\frac{t}{4M}\right] \right)}{4M^2 \sqrt{-4 + \frac{2r}{M}}}$$